AD-A107 608 NAVAL SURFACE WEAPONS CENTER SILVER SPRING MD F/G 7/4
DECONVOLUTION OF CHROMATOGRAMS.(U)
SEP 81 D W CARLSON, E G POWELL
UNCLASSIFIED NSWC/TR-81-307 NL



END
DATE FILMED
1-82
DTIC

NSWC TR 81-307

DECONVOLUTION OF CHROMATOGRAMS

BY DORN W. CARLSON E. GORDON POWELL

RESEARCH AND TECHNOLOGY DEPARTMENT

14 SEPTEMBER 1981

NAVAL SURFACE WEAPONS CENTER

Dahlgren, Virginia 22448 • Silver Spring, Maryland 20910

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE	READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER: NSWC TR 81-307 — 2. GOVT ACCESSION NO.: AD-A107 606	3. RECIPIENT'S CATALOG NUMBER:
4. TITLE (and Subtitle): Deconvolution of Chromatograms	5. TYPE OF REPORT & PERIOD COVERED: Final; Fiscal Year 1981
	6. PERFORMING ORG. REPORT NUMBER:
7. AUTHOR(s): Dorn W. Carlson and E. Gordon Powell	8. CONTRACT OR GRANT NUMBER(s):
9. PERFORMING ORGANIZATION NAME AND ADDRESS: Naval Surface Weapons Center, White Oak, Silver Spring, Maryland 20910	10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS: 61152N, ZR000-01-01, ZR01301, 2R01AA230
11. CONTROLLING OFFICE NAME AND ADDRESS:	12. REPORT DATE: 14 September 1981
	13. NUMBER OF PAGES: 28
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office):	15. SECURITY CLASS. (of this report): UNCLASSIFIED
	15a. DECLASSIFICATION/DOWNGRADING SCHEDULE:

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

A modified version will be submitted for publication in the Journal of Chromatography.

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Chromatography
Deconvolution
Computer-Aided Chrcmatography
BASIC

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A simple deconvolution algorithm based on a special case of the Scofield iterative technique was developed for use in deconvolving gas and liquid chromatograms. This algorithm required the use of only very limited computer equipment, such as a BASIC programmable integrator.

This technique can in certain cases sharpen chromatographic peaks mathematically to the extent that nonresolved shoulders or

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE
S/N 0102-LF-014-6601

UNCLASSIFIED
SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

cont.

doublets can be resolved and quantified.7

FOREWORD

The Energetic Materials Division has an ongoing program investigating double-base propellant combustion mechanisms. This effort is supported by the NAVSEA contract N000248WR10192 element 61153N and by NSWC internal Independent Research funding, 1R01AA230.

In support of these investigations the problem arose of characterizing two closely spaced chromatographic peaks. This paper reports the use of the Scofield unfolding algorithm as a method of mathematical deconvolution. Due to the Scofield algorithms property of preserving the sign of intermediate results a superior end result is achieved. The Scofield algorithm was initially developed at the U.S. Naval Radiological Defense Laboratory and was further applied here at the Naval Surface Weapons Center to various radiological programs. This report represents a further application of this remarkable algorithm. No endorsement of commercial products mentioned in this report is implied.

The authors express their appreciation to Cindy McPherson for assistance in preparation of the graphics, and to representatives of Waters Associate and Perkin-Elmer Corporation for helpful comments.

G. L. MacKenzie

G. L. MACKENZIE
By direction



CONTENTS

ILLUSTRATIONS

INTRODUCTION

Satisfactory resolution of chromatographic peaks is necessary for both qualitative and quantitative analysis. Despite the wide variety of possible combinations of detectors, columns and solvents, it is common to encounter poorly resolved or unresolved peaks. Various methods of mathematical analysis of the curves can then be applied as a last resort.

One such method models the chromatogram as a summation of gaussian curves or modified gaussian curves by least squares techniques[1]. This method can be useful but suffers to the extent that it does not handle non-gaussian or tailing peaks. Another broad class of techniques is digital filtering in one or more of its disguises. For example, if the shape of an unblended peak is shown it is assumed to represent the instrumental response function. An "optimal" filter can then be accomplished by dividing the Fourier transform of the unresolved data by the Fourier Transform of the response function[2]. The advent of the Fast Fourier Transform makes this elegant technique efficient and readily applied to automated chromatographs. However the optimality of the Fourier filtering process depends upon the fundamental assumption of mathematical linearity. Although the physical peaks do add linearly, it is also true that a negative amount of matter is a physical impossibility (i.e. a non-linear constraint). Thus we have the a priori knowledge that the solution to the resolution problem can only have positive peaks when the specimen affects the detector positively and only negative peaks when the specimen affects the detector negatively. The Fourier Transform admits the possibility of solutions of the opposite sign, consequently optimality can not be proven.

---

[1]Scott, Charles D., Chilcote, Dennis D., Pitt, W. Wilson, "Method for Resolving and Measuring Overlapping Chromatographic Peaks by use of an On-Line Computer with Limited Storage Capacity", Clinical Chemistry 16 (8) 637-42 (1970).

[2]Rabiner, Lawrence R. and Gold, Bernard, Theory and Application of Digital Signal Processing, Prentics Hall, Inc., Englewood Cliffs, NJ (1975).

Although no claims to optimally are known, the iterative unfolding technique of N. E. Scofield[3] admits only positive solutions, is easy to implement, and, as will be shown here, gives excellent results. The Scofield algorithm was designed to solve the more complicated problem of two-dimensional unfolding while we require here one dimensional unfolding otherwise known as deconvolution. The two dimensional problem occurs when the response function itself it not constant. While this is true of chromatograms in the most general case we limit ourselves here to deconvolving only a portion of a chromatogram.

This allows us to use the degenerate one-dimensional case of the Scofield algorithm which has the additional advantages that it can be programmed into any microcomputer which possesses the following capabilities: (1) ability to store digitized detector output data as a function of time in an array of a length equal to the number of equidistant time-slices taken, (2) minimal BASIC logic (IF/THEN, FOR/NEXT, GOTO, LET, +, -, /, *) or equivalent, and (3) enough RAM space to contain four detector output arrays and the BASIC program itself (about 50 logical lines for one no-frills version which included data acquisition).

In our laboratory this algorithm has been programmed and run on a Sigma 115 GC System console (Perkin-Elmer, Norwalk, CT) and on a Model 720 HPLC System Controller (Waters Assoc., Milford, MA). Examples of the BASIC programs which implement the algorithm for these two instruments are included in the Appendix.

DESCRIPTION OF ALGORITHM

The following variables are defined in the deconvolution algorithm:

C(x) = an array which contains baseline corrected time-slice data for the sample chromatogram that is to be deconvolved.

R(x) = an array which contains baseline corrected time-slice data for a reference chromatogram of a single pure substance, such as one component of the sample mixture.

N = the length of C(x); the number of time-slices taken to fill C(x).

N1 = the length of R(x); the number of time-slices taken to fill R(x). N1 need not be equal to N, but the slice width used for C(x) and R(x) should be the same.

---

[3]Scofield, N. E., "Iterative Unfolding", Applications of Computers to Nuclear and Radiochemistry, National Academy of Science Monograph NAS-NS-3107, G. D. O'Kelley, ed. (1962).

x1 = an integer such that $1 \leq x1 \leq N1$, and $R(x1) \leq R(x)$ for all $1 \leq x \leq N1$ (i.e., x1 is the number of the time-slice which contains the "peak" of the reference chromatogram).

N1(x) = an array of length N which is filled by the computer with the time-slice data for the deconvolved chromatogram N1(x) is initially set equal to C(x), and is updated with each iteration of the deconvolution algorithm.

C1(x) = an array of length N which is used by the computer to store intermediate values during each iteration.

The algorithm defines the following initial conditions:

$N1(x) = C(x)$ for all $1 \leq x \leq N$.

Each iteration of the deconvolution algorithm updates N(x) by the following equations:

$$C1(x) = \sum_{i=Y1}^{Y2} R(x1+i) * N(x+i) \quad \text{for all } 1 \leq x \leq N$$

(previous iteration)

Where: $Y1 = \max((1-x), (1-x1))$

$Y2 = \min((N-x), (N1-x1))$

$$N(x) = (C(x)/C1(x)) * N(x) \quad \text{for all } 1 \leq x \leq N$$

(previous iteration)

The vector N(x) can be renormalized after each iteration to conserve area or peak height. The deconvolved chromatogram N(x) shows the most marked improvement over C(x) after one or two iterations. After this, further improvement occurs approximately as the logarithm of the number of iterations performed. With each iteration, the chromatogram becomes increasingly noisy, so that it is advantageous to iterate as few times as possible to obtain a sufficiently sharpened chromatogram.

EXAMPLE

Figure 1 shows a simple chromatogram which was deconvolved on a Waters Model 720 HPLC System Controller. The sample was a mixture of 38% by weight 2-nitrodiphenylamine (2NDPA) and 62% diethylphthalate (DEP). The mixture was separated on a Waters brand silica radial compression column. The mobile phase used was 3/1, 2,2,4-trimethylpentane/tetrahydrofuran. A 254nm UV absorbance detector was used.

The original chromatogram is shown in Figure 1. The major peak eluting at 2.63 minutes is 2NDPA; the shoulder is DEP.

Figure 2 shows an expanded segment of this same chromatogram. It is composed of 38 data points taken at 0.03 minute intervals. The deconvolution process was performed on these data. The deconvolved chromatograms are shown in Figures 3 and 4 after undergoing two and ten iterations of the algorithm respectively. For convenience, after each iteration the data were normalized so that the maximum peak height was exactly full scale on the chart. After ten iterations, the main peaks are sharp enough that the DEP is almost baseline resolved. The area of the two peaks (times response factors) give an analysis for the sample of 35% 2NDPA, 65% DEP. This is in good agreement with the sample's actual weight make-up of 38% 2NDPA, 62% DEP.

In Figure 4 a shoulder can be seen on the DEP peak which was totally invisible in the non-deconvolved chromatogram. With further iterations, this shoulder too would become a baseline resolved peak. This fact points out a limitation on the power of the iterative algorithm: an energy bump in the chromatogram, whether signal peak or just random noise, is sharpened with each iteration. Without further prior knowledge the reality of such features is an epistemological question not addressed here. The main utility of this deconvolution technique is in the resolution of known peaks. It is at best a very risky tool for detecting previously unseen peaks. One can however construe such low level peaks as a hint directing future work.

OPERATOR: DORN CARLSON

SAMPLE NAME: SAMPLE (2-NDPA PLUS DIETHYL PHTHALATE) ACTUAL LC TRACE

SAMPLE POS 01, INJ VOLUME 0003, NO OF INJ 1, RUN TIME 00:04, SM
WISP CODES GENERATED: 63-01UL

MAY. 26, 1981 13:31:25 CHART 1.00 CM/MIN FLOW 2.00 ML/MIN
RUN #9 CHNL #0
COLUMN SOLVENT OPR ID: 5

EXTERNAL STANDARD QUANTITATION

PEAK#	AMOUNT	RT	EXP RT	AREA	RF
	36604.10000	2.63		36604289 L	0.000000E0
TOTAL	36604.10000				

PEAK#	AMOUNT	RT	%
	36604.10000	2.63	100.00
TOTAL	36604.10000		100.00

Figure 1 Liquid Chromatogram of 2NDPA/DEP Mixture



Figure 2 Digitized Data of Segment of Chromatogram in Figure 1



Figure 3 Deconvolved Chromatogram (two iterations)



Figure 4 Deconvolved Chromatogram (Ten Iterations)

APPENDIX A - "PROGRAM DOCUMENTATION"

The following BASIC programs are deconvolution programs written to operate on the Waters Associates programmable integrator the Model 720/730, and on the Perkin-Elmer Sigma 115 instrument. Figures A-1 and A-2 are notated excerpts of the programs which are specifically concerned with the deconvolution iteration. Figures A-3 and A-4 are complete programs which include some aspects of data acquisition, baseline corrections, and replotting.

```
7080 CPRINT "DECONVOLUTING...."
7100 CPRINT
7102 FOR X=1 TO N              ' Set up initial conditions, N(x) = C(x)
7105 N(X)=C(X)
7108 NEXT X
7120 CPRINT
7140 N0=0
7160 D=CHR(13)
7180 CPRINT "NO. OF ITERATIONS = ",N0 ,
7185 D=CHR(7)
7190 IF N0=10 THEN 8000        ' Do a total of ten iterations
7260 REM "DECONVOLUTION"
7280 X=0                       ' Deconvolve each point on curve, x
7290 X=X+1                     ' from x = 1 to x = N
7300 C1(X)=0                   ' Set C1(x), intermediate array, = to Ø
7320 Y1=1-X
7340 IF X1<X THEN Y1=1-X1      ' Y1 = max( 1-x, 1-X1 )
7360 Y2=N-X
7380 IF (N1-X1)<(N-X) THEN Y2=N1-X1        ' Y2 = min( N-x, N1-X1 )
7400 I=Y1-1
7410 I=I+1                     ' Do summation from i = Y1 to i = Y2
7420 C1(X)=C1(X)+R(X1+I)*N(X+I)
7460 IF I<Y2 THEN 7410         ' Next i
7500 IF X<N THEN 7290          ' Next point on curve (next x)
7560 X=0
7570 M1=0
7580 X=X+1
7600 IF C1(X)=0 THEN 7660      ' Division by Ø excluded
7620 N(X)=N(X)*C(X)/C1(X)
7640 GOTO 7680
7660 N(X)=0
7680 IF N(X)>M1 THEN M1=N(X)   ' Find M1, the maximum height of curve
7700 IF X<N THEN 7580
7720 X=0
7740 X=X+1
7760 N(X)=INT(N(X)*100000/M1+.5)/1000     ' Divide each point by M1/1ØØ
7780 IF X<N THEN 7740                     ' (height normalization)
7800 N0=N0+1
7820 IF N0=2 THEN 7860
7840 GOTO 7160
```

Figure A-1 BASIC Subroutine Used to Deconvolve Chromatograms on Waters Model 720

```
360 LET N0=0
370 LET M=0
380 FOR X=1 TO N
390 IF M<N(X) THEN LET M=N(X)          ' Find M, maximum height of curve
400 NEXT X
410 FOR X=1 TO N
420 LET N(X)=N(X)*70/M                 ' Divide each point on curve
430 NEXT X                             ' by M/70 to normalize
460 FOR X=1 TO INT(N/2)
470 LET A1=INT(N(2*X-1))               ' Lines 460-530 trace out curve
480 LET A2=INT(N(2*X))
490 IF A1=A2 THEN PRINT TAB(A1);"*"
500 IF A1<A2 THEN PRINT TAB(A1);"'";TAB(A2);"."
510 IF A1>A2 THEN PRINT TAB(A2);".";TAB(A1);"'"
520 NEXT X
530 IF N/2>INT(N/2) THEN PRINT TAB(INT(N(N)));"'"
540 PRINT "ITERATION NO. ";N0
550 LET N0=N0+1
560 IF N0=11 THEN 1000                 ' Do ten iterations
570 FOR X=1 TO N
580 LET D(X)=0                         ' D(x) corresponds to C1(x) in
600 LET Y1=1-X                         ' the algorithm
610 IF X1<X THEN LET Y1=1-X1           ' Y1 = max( 1-x, 1-X1 )
620 LET Y2=N-X
630 IF (N1-X1)<(N-X) THEN LET Y2=N1-X1        ' Y2 = min( N-x, N1-X1 )
640 FOR I=Y1 TO Y2                     ' Do summation from i = Y1 to i = Y2
650 LET D(X)=D(X)+R(X1+I)*N(X+I)
660 NEXT I
670 NEXT X
690 FOR X=1 TO N
700 IF D(X)=0 THEN 730                 ' Division by Ø excluded
710 LET N(X)=N(X)*C(X)/D(X)
720 GOTO 740
730 LET N(X)=0
740 NEXT X
750 GOTO 370
1000 REM
```

Figure A-2 BASIC Subroutine Used to Deconvolve Chromatograms on Perkin-Elmer Sigma 115

```
 REM "DECONV.BAS  DECONVOLUTION  5/6/81"
: REM "DORN CARLSON"
10 DEBUG
20 CLEAR
30 D=BACKF
40 D=KILLF
50 D=PAGEF
60 D=MODULEF
70 D=ERASEF
4000 DIM R(100)
4020 DIM D(500)
4040 DIM C(500)
4060 DIM C1(500)
4080 DIM N(500)
5000 ERASE
5010 D=BACKF
5020 K1=0
5040 CPRINT "RESPONSE FUNCTION (REFERENCE PEAK)"
5060 GOTO 5120
5080 ERASE
5100 CPRINT "RAW CHROMATOGRAM"
5120 CPRINT
5140 CPRINT "1 = COLLECT DATA AS SAMPLE IS RUN"
5150 X=0
5160 CPRINT "2 = LOAD DATA FROM TAPE"
5165 IF K1=1 THEN 5180
5170 CPRINT "3 = USE SAME RESPONSE FUNCTION"
5180 CINPUT ,X
5200 D=LIN(7)
5220 IF X=2 THEN 9000
5230 IF X=3 THEN 6100
5235 CPRINT
5240 CPRINT "INITIAL TIME, FINAL TIME, INTERVAL"
5260 CINPUT ,TAB(2),R1,TAB(16),R2,TAB(28),R3
5280 CPRINT
5290 CPRINT
5300 CPRINT "HIT <ENTER> AFTER RUN STARTS"
5320 IF BACKF=0 THEN 5320
5340 TO PPI ASK RUN_TIME=CON(Q)
5360 Q1=SYSTIM
5380 Q2=INT(Q1/100)-40*INT(Q1/10000)+INT((Q1-100*INT(Q1/100))/.6+.5)/10
0
5400 R4=Q2-Q-.01
5401 REM "R4=DIFF. BTN. RUN TIME & SYSTIM"
5410 N=1
5420 D=CHR(7)
5430 R=R1
5440 Q1=SYSTIM
5450 Q2=INT(Q1/100)-40*INT(Q1/10000)+INT((Q1-100*INT(Q1/100))/.6+.5)/
```

Figure A-3 Complete BASIC Deconvolution Program for Waters Instrument

```
[illegible]-P4
5440 [illegible]F Q2<R THEN 5440
5500 [illegible] PPI ASK INPUT_LEVEL=CON(Q)
5520 [illegible]=0
5540 [illegible]
5560 [illegible]
5580 [illegible] THEN 56[illegible]0
5600 GOTO 5440
5620 D=CHR(7)
5640 CPRINT "DATA COLLECTION COMPLETE"
5660 CPRINT "CORRECTING FOR BASELINE DRIFT"
5680 N=N-1
5700 A=C(1)
5720 B=C(N)
5740 X=0
5760 X=X+1
5780 C(X)=C(X)-((X-1)*B+(N-X)*A)/(N-1)
5790 IF C(X)<0 THEN C(X)=0
5800 IF X<N THEN 5760
5820 IF K1=1 THEN 6040
5840 S1=R1
5860 S2=R2
5880 S3=R3
5900 N1=N
5910 M1=0
5920 X=0
5940 X=X+1
5960 R(X)=C(X)
5980 IF R(X)>M1 THEN X1=X
6000 IF R(X)>M1 THEN M1=R(X)
6020 IF X<N THEN 5940
6030 GOTO 6060
6040 X=0
6045 M1=1
6048 X=X+1
6050 IF C(X)>M1 THEN M1=C(X)
6052 IF X<N THEN 6048
6055 X=0
6056 X=X+1
6057 C(X)=INT(C(X)*100000/M1+.5)/1000
6058 IF X<N THEN 6056
6060 GOSUB 6380
6100 IF K1=1 THEN 6160
6120 K1=1
6140 GOTO 5080
```

Figure A-3 (Continued)

```
6160 CPRINT "RESPONSE FUNCTION",TAB(22),"TRACE DATA"
6200 CPRINT
6220 CPRINT N1," POINTS",TAB(22),N," POINTS"
6240 CPRINT S1," START",TAB(22),R1," START"
6260 CPRINT S3," INTERVAL"
6280 CPRINT X1," PEAK MAX"
6320 CPRINT
6340 GOSUB 6380
6360 GOTO 6440
6380 CPRINT "HIT <ENTER> TO CONTINUE"
6400 IF BACKF=0 THEN 6400
6410 ERASE
6420 RETURN
6440 CPRINT "1 = STORE RESPONSE FUNCTION ON TAPE? " ,
6460 X=0
6480 CINPUT X
6500 CPRINT
6510 IF X=0 THEN 7040
6520 REM *****LINES_6720_TO_7020_NOT_IN_
7040 CPRINT
7060 GOSUB 6380
7080 CPRINT "DEC
7100 CPRINT
7102 FOR X=1 TO
7105 N(X)=C(X)
7108 NEXT X
7120 CPRINT
7140 N0=0
7160 D=CHR(13)
7180 CPRINT "NO. OF ITERATIONS = ",N0 ,
7185 D=CHR(7)
7190 IF N0=10 THEN 8000
7260 REM "DECONVOLUTION"
7280 X=0
7290 X=X+1
7300 C1(X)=0
7320 Y1=1-X
7340 IF X1<X THEN Y1=1-X1
7360 Y2=N-X
7380 IF (N1-X1)<(N-X) THEN Y2=N1-X1
7400 I=Y1-1
7410 I=I+1
7420 C1(X)=C1(X)+R(X1+I)*N(X+I)
7460 IF I<Y2 THEN 7410
7500 IF X<N THEN 7290
7560 X=0
7570 M1=0
7580 X=X+1
7600 IF C1(X)=0 THEN 7660
7620 N(X)=N(X)*C(X)/C1(X)
```

Figure A-3 (Continued)

```
7640 GOTO 7680
7660 N(X)=0
7680 IF N(X)>M1 THEN M1=N(X)
7700 IF X<N THEN 7580
7720 X=0
7740 X=X+1
7760 N(X)=INT(N(X)*100000/M1+.5)/1000
7780 IF X<N THEN 7740
7800 N0=N0+1
7820 IF N0=2 THEN 7860
7840 GOTO 7[illegible]0
7860 X=0
7880 X=X+1
7900 D(X)=N(X)
7920 IF X<N THEN 7880
7940 GOTO 7160
8000 CPRINT
8020 CPRINT "STORING DATA IN GRADIENT TABLE, PUMP"
8040 CPRINT "SET 13"
8050 CPRINT
8060 CPRINT "RAW DATA           %A"
8080 CPRINT "2 ITERATIONS       %B"
8100 CPRINT "10 ITERATIONS      %C"
8120 TO PUMP SET EDIT_SET=CON(13)
8140 TO PUMP SET ERASE_GD_TABLE=1
8160 TO PUMP SET TOTAL_FLOW=CON(0,C(1),D(1),N(1))
8180 R=R1
8200 N2=1
8220 TO PUMP SET GRADIENT_TABLE=CON(R,0,C(N2),D(N2),N(N2),"06")
8240 R=R+R3
8260 N2=N2+1
8280 IF N2>N THEN 8320
8300 GOTO 8220
8320 D=CHR(7)
8330 X=0
8340 CPRINT "1 = STORE DATA ON TAPE? " ,
8350 D=BACKF
8360 CINPUT ,X
8380 IF X=0 THEN 8580
8400 CPRINT "NAME OF FILE (" ,
8402 D=CHR(120)+CHR(120)+CHR(120)+CHR(120)+CHR(120)+CHR(120)
8404 D=CHR(46)+CHR(120)+CHR(120)+CHR(120)+CHR(41)+CHR(32)
8420 CINPUT ,STRING
  40 CPRINT
  50 CPRINT "(SAVE PUMP SET 13 ONLY!)"
  30 CPRINT "HIT <ENTER> WHEN TAPE IS READY"
  00 IF BACKF=0 THEN 8500
```

Figure A-3 (Continued)

```
8520 TSAVE PARAMETER, STRING
8530 T=TERRF
8540 IF T=1 THEN CPRINT "TAPE ERROR"
8545 IF T=1 THEN 8340
8560 CPRINT "COMPLETE"
8565 D=BACKF
8580 GOSUB 6380
8600 GOTO 5000
9000 CPRINT "THIS PART OF PROGRAM NOT COMPLETED"
9010 GOTO 5140
```

Figure A-3 (Continued)

```
 90 DIM C(200),N(200),D(200),R(200)
100 PRINT "PP OR RS";
110 INPUT B$
120 OPEN B$
130 PRINT "REF SLICE FILE, SMP SLICE FILE";
140 INPUT R1,S1
150 FOR X=1 TO 200
160 GSLICE R1,X,A,T
170 GSLICE S1,X,A1,T
180 IF A>-1 THEN LET R(X)=A/1000
182 IF A>-1 THEN LET N1=X
190 IF A1>-1 THEN LET C(X)=A1/1000
192 IF A1>-1 THEN LET N=X
200 IF A+A1=-2 THEN 210
205 NEXT X
210 PRINT "CORRECTING BASELINE DRIFT"
220 LET P=R(1)
222 LET M1=0
230 LET Q=R(N1)
240 FOR X=1 TO N1
250 LET R(X)=R(X)-((X-1)*Q+(N1-X)*P)/(N1-1)
260 IF R(X)<0 THEN LET R(X)=0
270 IF R(X)>M1 THEN LET X1=X
280 IF R(X)>M1 THEN LET M1=R(X)
290 NEXT X
300 LET P=C(1)
310 LET Q=C(N)
320 FOR X=1 TO N
330 LET C(X)=C(X)-((X-1)*Q+(N-X)*P)/(N-1)
340 IF C(X)<0 THEN LET C(X)=0
350 NEXT X
360 LET N0=0
362 FOR X=1 TO N
364 LET N(X)=C(X)
366 NEXT X
370 LET M=0
380 FOR X=1 TO N
390 IF M<N(X) THEN LET M=N(X)
400 NEXT X
410 FOR X=1 TO N
420 LET N(X)=N(X)*70/M
430 NEXT X
440 FOR X=1 TO INT(N/2)
 70 LET A1=INT(N(2*X-1))
 80 LET A2=INT(N(2*X))
```

Figure A-4 Complete BASIC Deconvolution Program for Perkin-Elmer Instrument

```
482 LET A1=A1+1
484 LET A2=A2+1
490 PRINT TAB(A1);"*"
500 PRINT TAB(A2);"*"
520 NEXT X
530 PRINT TAB(INT(N(N)+1));"*"
540 PRINT "ITERATION NO. ";N0
550 LET N0=N0+1
560 IF N0=11 THEN 1000
570 FOR X=1 TO N
580 LET D(X)=0
600 LET Y1=1-X
610 IF X1<X THEN LET Y1=1-X1
620 LET Y2=N-X
630 IF (N1-X1)<(N-X) THEN LET Y2=N1-X1
640 FOR I=Y1 TO Y2
650 LET D(X)=D(X)+R(X1+I)*N(X+I)
660 NEXT I
670 NEXT X
690 FOR X=1 TO N
700 IF D(X)=0 THEN 730
710 LET N(X)=N(X)*C(X)/D(X)
720 GOTO 740
730 LET N(X)=0
740 NEXT X
750 GOTO 370
1000 REM
9999  END
OK
```

Figure A-4 (Continued)

DISTRIBUTION

	Copies
Commander Naval Ordnance Station Attn: Code 3033	1
Code 3031	1
TDT1TC	2
Indian Head, MD 20640	
Chief of Naval Research Attn: ONR-473	2
800 North Quincy St. Arlington, VA 22217	
Defense Technical Information Center Cameron Station Alexandria, VA 22314	12
Commander Naval Research Laboratory Attn: Dr. B. Kinzig	1
Dr. A. Harvey	1
Dr. J. Schnur	1
Washington, DC 20375	
Commander Naval Sea Systems Command Attn: SEA-03B	1
SEA-09G32	2
Naval Sea Systems Command Headquarters Washington, DC 20362	
Johns Hopkins University Applied Physics Laboratory Chemical Propulsion Information Agency Attn: Mr. R. Brown	1
Mr. H. Hoffman	1
Laurel, MD 20810	
Library of Congress Washington, DC 20540 Attn: Gift and Exchange Division	4


DATE
TIME